CS61A Notes – Week 3: Applicative vs. normal, recursive vs. iterative processes, orders of growth (solutions)
Applicative vs. Normal Order
QUESTIONS

1. Above, applicative order was more efficient. Define a procedure where normal order is more efficient.
Anything where not evaluating the arguments will save time works. Most trivially,

(define (f x) 3) ;; a function that always returns 3

When you call (f (fib 10000)), applicative order would choke, but normal order would just happily drop (fib 10000) and just return 3.

2. Evaluate this expression using both applicative and normal order: (square (random x)). Will you get the same result from both? Why or why not?

Unless you’re lucky, the result will be quite different. Expanding to normal order, you have (* (random x) (random x)), and the two separate calls to random will probably return different values.

3. Consider a magical function count that takes in no arguments, and each time it is invoked, it returns 1 more than it did before, starting with 1. Therefore, (+ (count) (count)) will return 3. Evaluate (square (square (count))) with both applicative and normal order; explain your result.

For applicative order, (count) is only called once – returns 1 – and is squared twice. So you have (square (square 1)), which evaluates to 1.

For normal order, (count) is called FOUR times:

(* (square (count)) (square (count))) =>

(* (* (count) (count)) (* (count) (count))) =>

(* (* 1 2) (* 3 4)) =>

24
Recursive vs. Iterative Processes
QUESTIONS: Will the following generate a recursive or iterative process?
1. (define (foo x)

     (* (- (+ (/ x 3) 4) 6) 2))
It’s not recursive, so it’s pretty pointless to ask what kind of process it generates.
2. (define (foo x)

    (if (= x 0) 0 (+ x (foo (- x 1))))
Recursive
3. (define (helper 1 x)

    (if (= x 0) 1 (helper1 (- x 1)))) => Iterative
(define (helper2 x)

    (if (= x 0) 1 (+ 1 (helper2 (- x 1))))) => Recursive
a. (define (bar x)

    (if (even? x) (helper1 (- x 1)) (helper1 (- x 2))))

Iterative
b. (define (bar x)

    (if (even? x) (helper2 (- x 1)) (helper2 (- x 2))))

Recursive
c. (define (bar x)

    (if (= x 0) (helper2 x) (helper1 x)))

Iterative (when x is 0, (helper2 0) returns immediately)
d. (define (bar x)

    (if (= x 0) (helper1 x) (helper2 x)))

Recursive
e. (define (bar x)
    (cond ((= x 0) 1)
          ((= (helper2 x) 3) 5)

          (else (helper1 x))))
Recursive
f. (define (bar x)
    (helper2 (helper1 x)))

Recursive

Yoshimi Battles the Recursive Robots, Pt. 2
QUESTIONS

1. Consider the subset-sum problem: you are given a sentence of integers and a number k. Is there a subset of the sentence that add up to k? For example,
(subset-sum ‘(2 4 7 3) 5) => #t, since 2+3=5
(subset-sum ‘(1 9 5 7 3) 2) => #f
(define (subset-sum? nums k)

    (cond ((= k 0) #t)

          ((empty? nums) #f)

          ((< k 0) #f)

          (else (or (subset-sum? (bf nums) k)

                (subset-sum? (bf nums) (- k (first nums)))))))

2. There is something called a “falling factorial”. (falling n k) means that k consecutive numbers should be multiplied together, starting from n and working downward. For example, (falling 7 3) means 7 * 6 * 5. Write the procedure falling that generates an iterative process.

(define (falling b n)


    (define (helper b n ans)
        (if (= n 1)

      (* b ans)

      (helper (- b 1) (- n 1) (* b ans))))

    (helper b n 1))
3. Write a version of (expt base power) that works with negative powers as well.

(define (expt base power)

    (cond ((= power 0) 1)

          ((> power 0) (* base (expt base (- power 1))))

          (else (/ (expt base (+ power 1)) base))))
4. Implement (ab+c a b c) that takes in values a, b, c and returns ( + (* a b) c). However, you cannot use *. Make it a recursive process.

(define (ab+c a b c)

    (if (= b 0)

        c

        (+ a (ab+c a (- b 1) c))))

Yes, this assumes b is positive.  So sue me.  What should you do if b is negative?
5. Implement (ab+c a b c) as an iterative process. Don’t define helper procedures.
(define (ab+c a b c)

    (if (= b 0)

        c

        (ab+c a (- b 1) (+ c a))))
Orders of Growth
QUESTIONS: What is the order of growth in time for:
1. (define (fact x)

    (if (= x 0)

        1

        (* x (fact (- x 1)))))
Θ(n), since we subtract 1 from x each time
2. (define (fact-iter x answer)

    (if (= x 0)

        answer

        (fact-iter (- x 1) (* answer x))))
Θ(n)
3. (define (sum-of-facts x n)

    (if (= n 0)

        0

        (+ (fact x) (sum-of-facts x (- n 1)))))
Θ(xn), since we call sum-of-facts n times, and each time we have to calculate (fact x)
4. (define (fib n)

    (if (<= n 1)

        1

        (+ (fib (- n 1)) (fib (- n 2)))))
Θ(2n), since we make two recursive calls each time (draw out the recursion tree and convince yourself)
5. (define (square n)
    (cond ((= n 0) 0)

          ((even? n) (* (square (quotient n 2)) 4))

          (else (+ (square (- n 1)) (- (+ n n) 1)))))

Θ(log n); we cut down the input size by half each time it’s even. When it’s odd, we make one extra recursive call, but then, once we do (- n 1), it’s even again, and we get to cut it in half.

